<commit_before>{ config, lib, pkgs, ... }:

with lib;

let
  cfg = config.services.xserver.gdk-pixbuf;

  # Get packages to generate the cache for. We always include gdk-pixbuf.
  effectivePackages = unique ([pkgs.gdk-pixbuf] ++ cfg.modulePackages);

  # Generate the cache file by running gdk-pixbuf-query-loaders for each
  # package and concatenating the results.
  loadersCache = pkgs.runCommand "gdk-pixbuf-loaders.cache" { preferLocalBuild = true; } ''
    (
      for package in ${concatStringsSep " " effectivePackages}; do
        module_dir="$package/${pkgs.gdk-pixbuf.moduleDir}"
        if [[ ! -d $module_dir ]]; then
          echo "Warning (services.xserver.gdk-pixbuf): missing module directory $module_dir" 1>&2
          continue
        fi
        GDK_PIXBUF_MODULEDIR="$module_dir" \
          ${pkgs.stdenv.hostPlatform.emulator pkgs.buildPackages} ${pkgs.gdk-pixbuf.dev}/bin/gdk-pixbuf-query-loaders
      done
    ) > "$out"
  '';
in

{
  options = {
    services.xserver.gdk-pixbuf.modulePackages = mkOption {
      type = types.listOf types.package;
      default = [ ];
      description = "Packages providing GDK-Pixbuf modules, for cache generation.";
    };
  };

  # If there is any package configured in modulePackages, we generate the
  # loaders.cache based on that and set the environment variable
  # GDK_PIXBUF_MODULE_FILE to point to it.
  config = mkIf (cfg.modulePackages != [] || pkgs.stdenv.hostPlatform != pkgs.stdenv.buildPlatform) {
    environment.variables = {
      GDK_PIXBUF_MODULE_FILE = "${loadersCache}";
    };
  };
}<commit_msg>nixos/gdk-pixbuf.nix: don’t set GDK_PIXBUF_MODULE_FILE in cross<commit_after>{ config, lib, pkgs, ... }:

with lib;

let
  cfg = config.services.xserver.gdk-pixbuf;

  # Get packages to generate the cache for. We always include gdk-pixbuf.
  effectivePackages = unique ([pkgs.gdk-pixbuf] ++ cfg.modulePackages);

  # Generate the cache file by running gdk-pixbuf-query-loaders for each
  # package and concatenating the results.
  loadersCache = pkgs.runCommand "gdk-pixbuf-loaders.cache" { preferLocalBuild = true; } ''
    (
      for package in ${concatStringsSep " " effectivePackages}; do
        module_dir="$package/${pkgs.gdk-pixbuf.moduleDir}"
        if [[ ! -d $module_dir ]]; then
          echo "Warning (services.xserver.gdk-pixbuf): missing module directory $module_dir" 1>&2
          continue
        fi
        GDK_PIXBUF_MODULEDIR="$module_dir" \
          ${pkgs.stdenv.hostPlatform.emulator pkgs.buildPackages} ${pkgs.gdk-pixbuf.dev}/bin/gdk-pixbuf-query-loaders
      done
    ) > "$out"
  '';
in

{
  options = {
    services.xserver.gdk-pixbuf.modulePackages = mkOption {
      type = types.listOf types.package;
      default = [ ];
      description = "Packages providing GDK-Pixbuf modules, for cache generation.";
    };
  };

  # If there is any package configured in modulePackages, we generate the
  # loaders.cache based on that and set the environment variable
  # GDK_PIXBUF_MODULE_FILE to point to it.
  config = mkIf (cfg.modulePackages != []) {
    environment.variables = {
      GDK_PIXBUF_MODULE_FILE = "${loadersCache}";
    };
  };
}
{'fraction_non_alphanumeric': 0.11751662971175167, 'fraction_numerical': 0.0012670256572695597, 'mean_word_length': 3.1281045751633987, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 7, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '7545857', 'n_tokens_mistral': 1043, 'n_tokens_neox': 1025, 'n_words': 241}